
pkgname=vpnc
pkgver=0.5.3.3
#_pkgver=0.5.3.svn550
pkgrel=3
pkgdesc="VPN client for cisco3000 VPN Concentrators"
url="http://www.unix-ag.uni-kl.de/~massar/vpnc/"
license=('GPL')
depends=('libgcrypt' 'openssl' 'iproute2')
makedepends=('subversion')
arch=('x86_64')
# get revision by cd to vpnc > svn info | awk '/^Revision:/ { print $2 }'
source=("vpnc::svn+http://svn.unix-ag.uni-kl.de/vpnc/trunk#revision=550"
        'vpnc.conf')
backup=(etc/vpnc/default.conf)
md5sums=('SKIP'
         'a3f4e0cc682f437e310a1c86ae198e45')

build() {
  cd $pkgname

  # Build hybrid support
  sed -i 's|^#OPENSSL|OPENSSL|g' Makefile

  make 
}

package() {
  cd $pkgname
  
  make DESTDIR=$pkgdir PREFIX=/usr install

  install -D -m644 $srcdir/vpnc.conf $pkgdir/etc/vpnc/default.conf
}


